Introduction to Bayesian inference using Rstan: practical 2 solutions

The aim of this practical is to provide some practical experience in writing Stan programmes and using the rstan package for posterior inference. Some sections require more experience with statistics than others. These are marked with an asterisk.

First load the rstan and jrRstan packages:

```
library("rstan")
library("jrRstan")
```

If you have enough RAM, set options to allow parallel computation:

```
rstan_options(auto_write = TRUE)
options(mc.cores = parallel::detectCores())
```

1 Binomial regression

In Section 5.2 we considered a generalised linear model in which we assumed a *Poisson* error distribution. In this section we will consider another regression problem, but this time we will use a *binomial* error distribution. This is called *binomial regression*. In the special case where a *logistic*¹ link function is utilised, the name *logistic regression* is often used. We will use a logistic link function in this example.

As you might expect, the proportion of people who suffer side effects from a drug typically depends on the dose of that drug they are given. The jrRstan package contains a data set called sideeffect which can be loaded via:

```
data(sideeffect)
head(sideeffect)

## dose n effects
## 1 0.9 46    17
## 2 1.1 72    22
## 3 1.8 118    52
## 4 2.3 96    58
## 5 3.0 84    56
## 6 3.3 53    43
```

For each of a number of doses (dose), the data set contains the number (n) of patients given a particular drug to treat a medical condition, and the number of those patients suffering from a particular side effect (effects).

Suppose that the aim is to model the number Y_i of the n_i patients receiving dose i that suffer side effects in terms of the dose \tilde{x}_i of the drug they are given. The model can be expressed as:

```
Y_i | n_i, p_i \sim \text{Bin}(n_i, p_i), independently for i = 1, 2, ..., N,
```

where here N = 7. We will mean centre the covariate, taking $x_i = \tilde{x}_i - \sum_{i=1}^N \tilde{x}_i / N$, and use a logistic link function to connect the linear predictor:

$$\eta_i = \beta_1 + \beta_2 x_i$$

¹ Other possibilities include the *probit* link or the *complementary log-log* link.

to the probability p_i that an individual receiving dose i suffers side effects. In other words:2

 $p_i = \frac{e^{\eta_i}}{1 + e^{\eta_i}}.$

or equivalently:3

$$\eta_i = \log\left(\frac{p_i}{1 - p_i}\right).$$

Our model contains two parameters: β_1 and β_2 . We will adopt the following prior:

$$\beta_1 \sim N(-0.27, 0.68^2)$$
, and, independently, $\beta_2 \sim N(0.47, 0.31^2)$.

• Write a Stan model to represent the model and prior above, remembering to mean-centre the covariate.

```
// Binomial regression model, saved in sideeffect.stan
data {
                          // Number doses
  int<lower=1> N;
  int<lower=1> K;
                          /* Number columns in design matrix,
                             i.e. number covariates + 1 */
 matrix[N, K] Xtilde;
                          // Design matrix
 int<lower=0> y[N];
                          // Binomial response
                          // Number patients (trials)
 int<lower=0> n[N];
                          // Means in beta priors
  real beta_mean[K];
  real<lower=0> beta_sd[K]; // Standard deviations in beta priors
}
transformed data {
  matrix[N, K] X; /* Centred design matrix; we could have
                    avoided this block by performing the
                     transformation in R. */
 X[,1] = Xtilde[,1];
 X[,2] = Xtilde[,2] - mean(Xtilde[,2]);
parameters {
  vector[K] beta;
model {
 vector[N] eta;
 eta = X * beta; // Fast matrix-vector calculation
  // Likelihood:
  y ~ binomial_logit(n, eta); /* Arithmetically stable form of
                                fast, vectorized sampling
                                statement */
 // Prior:
  beta ~ normal(beta_mean, beta_sd); /* Fast, vectorized sampling
                                       statement */
```

• Create a suitable data representation in R then compile and run the Stan programme.

```
## Set up data to pass to stan function in its "data" argument:
N = nrow(sideeffect) # Number doses, i.e. 7
                           # Number columns in design matrix
K = 2
Xtilde = matrix(1, N, K) # Design matrix
```

² This is called the *logistic* transformation

³ This is called the *logit* (or *inverse logistic*) transformation of p_i .

```
Xtilde[,2] = sideeffect$dose
y = sideeffect$effects  # Binomial response
n = sideeffect$n  # Number patients (trials)
beta_mean = c(-0.27, 0.47) # Means in beta priors
beta_sd = c(0.68, 0.31) # Standard deviations in beta priors
sideeffect_data = list(N=N, K=K, y=y, n=n, Xtilde=Xtilde,
                      beta_mean=beta_mean, beta_sd=beta_sd)
## Compile and run Stan programme:
chains = 4
iter = 2000
thin = 1
sideeffect_stan = stan("sideeffect.stan", data=sideeffect_data,
                        chains=chains, iter=iter, warmup=iter/2,
                        thin=thin)
```

• Check both numerical and graphical diagnostics.

```
## Numerical diagnostics:
print(sideeffect_stan)
## Inference for Stan model: sideeffect.
## 4 chains, each with iter=2000; warmup=1000; thin=1;
## post-warmup draws per chain=1000, total post-warmup draws=4000.
##
##
           mean se_mean sd 2.5%
                                      25% 50%
                                                     75% 97.5% n_eff
            0.30 0.00 0.1
                               0.10 0.23
## beta[1]
                                              0.30
                                                      0.36
                                                            0.49 3324
## beta[2] 0.75 0.00 0.1 0.55 0.68 0.75 0.82
                                                           0.95 3307
## lp__ -322.70 0.02 1.0 -325.44 -323.07 -322.38 -321.99 -321.72 1631
##
        Rhat
## beta[1] 1
## beta[2] 1
           1
## lp__
##
## Samples were drawn using NUTS(diag_e) at Tue Apr 4 22:56:29 2017.
## For each parameter, n_eff is a crude measure of effective sample size,
## and Rhat is the potential scale reduction factor on split chains (at
## convergence, Rhat=1).
## Graphical diagnostics:
sideeffect_stan_arr = as.array(sideeffect_stan)
diagnostics(sideeffect_stan_arr)
## NULL
## Rhat values all close to 1, n_eff large, graphical
## diagnostics give no cause for concern.
```

Random slope model for Oxboys data

In Section 5.3 we examined the 0xboys data where $Y_{i,j}$ and $x_{i,j}$ denoted the *j*-th measurement of height and age on boy *i*, where i = 1, ..., Iand j = 1, ..., J. We can load the data in R through:




Figure 1: Graphical diagnostics for binomial regression.

```
library(nlme)
data(Oxboys)
head(0xboys)
## Grouped Data: height ~ age | Subject
##
     Subject
                age height Occasion
## 1
          1 -1.0000 140.5
## 2
          1 -0.7479 143.4
                                  2
## 3
          1 -0.4630 144.8
                                  3
## 4
          1 -0.1643 147.1
                                  4
## 5
          1 -0.0027 147.7
                                  5
        1 0.2466 150.2
```

In lectures, we considered a random intercept model in which the intercept term in a regression of the height of a boy on his age could vary between individuals. In this section, we will consider an extension with both a random intercept and a random slope that can vary from one boy to the next. In other words we will fit a hierarchical model of the form:

$$Y_{i,j} = \alpha_0 + \alpha_{1,i} + (\beta_0 + \beta_{1,i})x_{i,j} + \epsilon_{i,j}$$
 where $\epsilon_{i,j}|\sigma^2 \sim N(0,\sigma^2)$ (1)

independently for i = 1, 2, ..., I, j = 1, 2, ..., J with:

$$\alpha_{1,i}|\sigma_{\alpha_1}^2 \sim \mathrm{N}(0,\sigma_{\alpha_1}^2) \quad \text{and} \quad \beta_{1,i}|\sigma_{\beta_1}^2 \sim \mathrm{N}(0,\sigma_{\beta_1}^2) \tag{2}$$

independently for i = 1, 2, ..., I. We now have two sets of random effects: the $\alpha_{1,i}$ and the $\beta_{1,i}$.

We complete our hierarchical model by specifying a prior for the parameters in the "top" level of the model (1):

$$\alpha_0 \sim N(m_{\alpha_0}, s_{\alpha_0}^2), \quad \beta_0 \sim N(m_{\beta_0}, s_{\beta_0}^2), \quad \sigma^2 \sim Gam(a_{\sigma^2}, b_{\sigma^2}),$$

and a hyperprior for the parameters in the "bottom" level of the model (2):

$$\sigma_{\alpha}^2 \sim \operatorname{Gam}(a_{\sigma_{\alpha}^2}, b_{\sigma_{\alpha}^2}), \quad \sigma_{\beta}^2 \sim \operatorname{Gam}(a_{\sigma_{\beta}^2}, b_{\sigma_{\beta}^2}).$$

As in Chapter 5, when fitting the model we will take:

$$m_{\alpha_0} = 140$$
, $s_{\alpha_0} = 3$, $m_{\beta_0} = 15$, $s_{\beta_0} = 7.5$
 $a_{\sigma^2} = 1.1$, $b_{\sigma^2} = 0.025$, $a_{\sigma^2_{\alpha}} = 1.1$, $b_{\sigma^2_{\alpha}} = 0.05$.

For the constants $a_{\sigma_{\beta}^2}$ and $b_{\sigma_{\beta}^2}$ in the extra hyperprior for σ_{β}^2 we will take:

$$a_{\sigma_{\beta}^2} = 1.1$$
, $b_{\sigma_{\beta}^2} = 0.1$.

• Extend the Stan programme for the random intercept model to additionally include a random slope.

```
// Random slope model, saved in oxboys_ranslope.stan
data {
 int<lower=1> N_subj;  // Number of subjects
int<lower=1> N;  // Sample size
real<lower=0> a_sigsq_alpha1; // Prior shape for sigsq_alpha1
  real<lower=0> b_sigsq_alpha1; // Prior rate for sigsq_alpha1
  real<lower=0> a_sigsq_beta1; // Prior shape for sigsq_beta1
  real<lower=0> b_sigsq_beta1; // Prior rate for sigsq_beta1
}
parameters {
  real alpha0;
  vector[N_subj] alpha1;
  real beta0;
 vector[N_subj] beta1;
  real<lower=0> sigsq;
  real<lower=0> sigsq_alpha1;
  real<lower=0> sigsq_beta1;
model {
 // Likelihood:
 vector[N] alpha;
 vector[N] beta;
```

```
alpha = alpha0 + alpha1[subj_label]; // Random intercept
  beta = beta0 + beta1[subj_label];  // Random slope
  y ~ normal(alpha + beta .* x, sqrt(sigsq));
  // Prior:
  alpha0 ~ normal(m_alpha0, s_alpha0);
  beta0 ~ normal(m_beta0, s_beta0);
  sigsq ~ gamma(a_sigsq, b_sigsq);
  alpha1 ~ normal(0, sqrt(sigsq_alpha1));
 beta1 ~ normal(0, sqrt(siqsq_beta1));
 // "Hyper-prior":
  sigsq_alpha1 ~ gamma(a_sigsq_alpha1, b_sigsq_alpha1);
  sigsq_beta1 ~ gamma(a_sigsq_beta1, b_sigsq_beta1);
}
```

• Create a suitable data representation in R then compile and run the Stan programme.

```
## Set up data to pass to stan function in its "data" argument:
y = 0xboys$height
                                      # Response var. (height)
x = 0xboys\$age
                                       # Covariate (age)
subj_label = as.numeric(Oxboys$Subject) # Subject label
N_subj = length(unique(subj_label)) # Number of subjects
## Data for random intercept model (from Chapter 5):
oxboys_ri_data = list(y=y, N=length(y), N_subj=N_subj,
            x=x, subj_label=subj_label,
            m_alpha0=140, s_alpha0=3, m_beta0=15, s_beta0=7.5,
            a_sigsq=1.1, b_sigsq=0.025,
             a_sigsq_alpha1=1.1, b_sigsq_alpha1=0.05)
## Data for random slope model:
oxboys_rs_data = oxboys_ri_data
oxboys_rs_data[["a_sigsq_beta1"]] = 1.1
oxboys_rs_data[["b_sigsq_beta1"]] = 0.1
## Compile and run Stan programme:
chains = 4
iter = 20000
thin = 10
oxboys_rs_stan = stan("oxboys_ranslope.stan", data=oxboys_rs_data,
                      chains=chains, iter=iter, warmup=iter/2,
                      thin=thin)
```

• Check both numerical and graphical diagnostics.

```
## Numerical diagnostics (here omitting the output for the
## random effects in alpha1 and beta1 to save space):
print(oxboys_rs_stan, pars=c("alpha1", "beta1"), include=FALSE)
## Inference for Stan model: oxboys_ranslope.
## 4 chains, each with iter=20000; warmup=10000; thin=10;
## post-warmup draws per chain=1000, total post-warmup draws=4000.
##
##
                                             25%
                                                            75%
                                                                 97 5%
                mean se_mean sd 2.5%
                                                     50%
## alpha0
               147.41 0.03 1.42 144.53 146.48 147.43 148.37 150.11
            6.54 0.01 0.37 5.81 6.31 6.54 6.78
## beta0
```

```
0.45
                          0.00 0.05
                                       0.36
                                                0.41
                                                      0.44 0.48
## sigsq
                                                                        0.55
                                               51.16
                 62.84
                                       37.83
                                                       60.86
                                                              71.76 100.17
## sigsq_alpha1
                          0.27 16.28
## sigsq_beta1
                  3.28
                          0.02 1.06
                                        1.80
                                                2.54
                                                        3.11
                                                                3.80
                                                                        5.84
## lp__
               -118.90
                          0.10 6.02 -131.55 -122.74 -118.51 -114.54 -108.33
               n_eff Rhat
##
## alpha0
                2637
## beta0
                2717
                        1
## sigsq
                3921
## sigsq_alpha1 3741
                        1
## sigsq_beta1
                3982
                        1
## lp__
                3861
                        1
##
## Samples were drawn using NUTS(diag_e) at Tue Apr 4 22:58:37 2017.
\#\# For each parameter, n_eff is a crude measure of effective sample size,
## and Rhat is the potential scale reduction factor on split chains (at
## convergence, Rhat=1).
# Graphical diagnostics:
oxboys_rs_stan_arr = as.array(oxboys_rs_stan)
diagnostics(oxboys_rs_stan_arr)
## NULL
## Rhat values all close to 1, n_eff large, graphical
## diagnostics give no cause for concern. However, note
## that this required a larger sample than the default
## of iter=2000.
```




Figure 2: First page of graphical diagnostics for random slope model.

• More difficult: Modify the Stan programmes for the random intercept model and random slope model to include a generated quantities block which computes the deviance of the model. (This will be similar to the code for the linear regression model in Chapter 3). Compute the DIC for each model and use it to decide which model is better according to this criterion.

```
// Random intercept model, saved in oxboys_ranintercept_dev.stan
functions {
  real deviance(vector y, vector x, int[] subj_label,
                real alpha0, vector alpha1, real beta0,
                real sigsq) {
    real dev;
   vector[num_elements(y)] alpha;
   vector[num_elements(y)] eta;
    alpha = alpha0 + alpha1[subj_label]; // Random intercept
    eta = alpha + beta0 * x; // Single matrix-vector calculation
    dev = (-2) * normal_lpdf(y | eta, sqrt(sigsq)); /*
                                  Vectorized form of the normal
                                  probability function */
    return dev;
 }
}
// ...
// Stan programme for random intercept model from Chapter 5
// ...
generated quantities {
 real dev; // Deviance
 dev = deviance(y, x, subj_label, alpha0, alpha1, beta0, sigsq);
}
// Random slope model, saved in oxboys_ranslope_dev.stan
functions {
  real deviance(vector y, vector x, int[] subj_label,
                real alpha0, vector alpha1, real beta0,
                vector beta1, real sigsq) {
    real dev;
    vector[num_elements(y)] alpha;
    vector[num_elements(y)] beta;
    vector[num_elements(y)] eta;
    alpha = alpha0 + alpha1[subj_label]; // Random intercept
    beta = beta0 + beta1[subj_label]; // Random slope
    eta = alpha + beta .* x; // Single matrix-vector calculation
    dev = (-2) * normal_lpdf(y | eta, sqrt(sigsq)); /*
                                  Vectorized form of the normal
                                  probability function */
    return dev;
  }
}
// Stan programme for random slope model above
// ...
generated quantities {
  real dev; // Deviance
 dev = deviance(y, x, subj_label, alpha0, alpha1, beta0,
                 beta1, sigsq);
```

```
## Compile and run the random intercept programme:
oxboys_ri_stan = stan("oxboys_ranintercept_dev.stan", data=oxboys_ri_data,
                      chains=chains, iter=iter, warmup=iter/2,
                      thin=thin)
## Compile and run the random slope programme:
oxboys_rs_stan = stan("oxboys_ranslope_dev.stan", data=oxboys_rs_data,
                      chains=chains, iter=iter, warmup=iter/2,
                      thin=thin)
## Compute the DIC for the random intercept model:
output = as.array(oxboys_ri_stan)
(D_bar = mean(output[,,"dev"])) # Posterior expectation of deviance
## [1] 792.3444
alpha0_bar = mean(output[,,"alpha0"])
alpha1_bar = apply(output[,,paste("alpha1[",1:26,"]", sep="")],
                   3, mean)
beta0_bar = mean(output[,,"beta0"])
sigsq_bar = mean(output[,,"sigsq"]) # Posterior mean of params
expose_stan_functions(oxboys_ri_stan) # Export deviance function
D_theta_bar = deviance(oxboys_ri_data$y, oxboys_ri_data$x,
                          oxboys_ri_data$subj_label, alpha0_bar,
                          alpha1_bar, beta0_bar, sigsq_bar)
                          # Deviance at posterior mean
(pD = D_bar - D_theta_bar) # Effective number of parameters
## [1] 27.94429
(DIC = D_bar + pD) # DIC
## [1] 820.2887
## Compute the DIC for the random slope model:
output = as.array(oxboys_rs_stan)
(D_bar = mean(output[,,"dev"])) # Posterior expectation of deviance
## [1] 471.7068
alpha0_bar = mean(output[,,"alpha0"])
alpha1_bar = apply(output[,,paste("alpha1[",1:26,"]", sep="")],
                   3, mean)
beta0_bar = mean(output[,,"beta0"])
betal_bar = apply(output[,,paste("beta1[",1:26,"]", sep="")],
                   3, mean)
sigsq_bar = mean(output[,,"sigsq"]) # Posterior mean of params
expose_stan_functions(oxboys_rs_stan) # Export deviance function
D_theta_bar = deviance(oxboys_rs_data$y, oxboys_rs_data$x,
                          oxboys_rs_data$subj_label, alpha0_bar,
                          alpha1_bar, beta0_bar, beta1_bar,
                          sigsq_bar)
                          # Deviance at posterior mean
(pD = D_bar - D_theta_bar) # Effective number of parameters
## [1] 51.95708
```

```
(DIC = D_bar + pD) # DIC

## [1] 523.6639

## The random slope model has the smaller DIC and so is
## to be preferred.
```

3 Hierarchical model for rat tumour data

In this section we will consider a well-known data set that is amenable to Bayesian hierarchical modelling. The data are available from the jrRstan package and concern the proportion of rats developing tumours in a number of laboratory studies. We can load the data via:

```
data(rats)
head(rats)

##  y  n
## 1 0 20
## 2 0 20
## 3 0 20
## 4 0 20
## 5 0 20
## 6 0 20
```

Our goal is to learn about the population probability of tumour amongst rats.

For study i, i = 1, ..., N, n_i denotes the total number of rats and Y_i denotes the number of rats who developed a tumour. To allow for the differences between studies in rats and experimental conditions, we let the probability of developing a tumour vary between studies and denote it by p_i for study i. We then assume the Y_i are independent binomial random variables given the study-specific probabilities p_i :

$$Y_i|n_i, p_i \sim \text{Bin}(n_i, p_i)$$
, independently for $i = 1, 2, ..., N$,

where here N=71. Suppose we initially thought the probabilities p_i might be around 0.1. If we were to learn that one probability p_i was larger than 0.1, this would cause us to revise upwards our "best guess" at the probabilities for other studies because we expect the p_i to be similar. To formalise this idea, we will assume the p_i are samples from some population distribution with common mean and variance. This is an example of a *Bayesian hierarchical model*. For convenience, we will reparameterise the binomial distributions so that instead of working with probabilities p_i we work with their logit transformations:

$$\theta_i = \log\left(\frac{p_i}{1 - p_i}\right)$$

and assume that

$$\theta_i | \mu, \sigma^2 \sim N(\mu, \sigma^2).$$

We allow the parameters μ and σ^2 to be unknown and give them prior distributions:

$$\mu \sim N(m_{\mu}, s_{\mu}^2)$$
 and $\sigma^2 \sim Gam(a_{\sigma^2}, b_{\sigma^2})$.

The mean μ can be thought of as the population value for the logit-probability of tumour amongst rats. The variance σ^2 quantifies the degree of variation in the population; the larger its value, the more heterogeneity we see in the probabilities between studies.

• Write a Stan model to represent the hierarchical model above.

```
// Hierarchical model for rat tumour data, saved in rats.stan
data {
 int<lower=0> N; // No. studies (sample size)
 int<lower=0> y[N]; // No. of rats with tumours
 int<lower=0> n[N]; // Total number of rats
                       // Prior mean for mu
 real m_mu;
  real<lower=0> s_mu; // Prior std. dev. for mu
  real<lower=0> a_sigsq; // Prior shape for sigsq
  real<lower=0> b_sigsg; // Prior rate for sigsg
}
parameters {
 vector[N] theta;
 real mu;
 real<lower=0> sigsq;
model {
 // Likelihood:
 y ~ binomial_logit(n, theta);
 // Prior:
 theta ~ normal(mu, sqrt(sigsq));
 // "Hyperprior":
 mu \sim normal(m_mu, s_mu);
 sigsq ~ gamma(a_sigsq, b_sigsq);
generated quantities {
 vector<lower=0, upper=1>[N] p; /* Study-specific probability
                                   of tumour */
  real<lower=0, upper=1> p_pop; /* "Population" probability
                                  of tumour */
  p = inv_logit(theta);
  p_pop = inv_logit(mu);
```

• Take

$$m_{\mu} = 0$$
, $s_{\mu}^2 = 2$, $a_{\sigma^2} = 1.1$, $b_{\sigma^2} = 1.3$.

Create a suitable data representation in R then compile and run the Stan programme.

```
iter = 2000
thin = 1
rats_stan = stan("rats.stan", data=rats_data,
                  chains=chains, iter=iter, warmup=iter/2,
                  thin=thin)
rats_reparam_stan = stan("rats_reparam.stan", data=rats_data,
                         chains=chains, iter=iter, warmup=iter/2,
                         thin=thin)
```

• Check both numerical and graphical diagnostics.

```
## Numerical diagnostics (here just printing a couple to save
## space):
print(rats_stan, pars=c("mu", "sigsq", "p_pop"))
## Inference for Stan model: rats.
## 4 chains, each with iter=2000; warmup=1000; thin=1;
## post-warmup draws per chain=1000, total post-warmup draws=4000.
##
##
         mean se_mean sd 2.5% 25% 50% 75% 97.5% n_eff Rhat
## mu -1.94 0.00 0.13 -2.20 -2.02 -1.94 -1.86 -1.71 1925 1.00
## sigsq 0.52 0.01 0.19 0.24 0.38 0.49 0.62 0.94
## p_pop 0.13 0.00 0.01 0.10 0.12 0.13 0.13 0.15 1912 1.00
## Samples were drawn using NUTS(diag_e) at Tue Apr 4 23:03:02 2017.
## For each parameter, n_eff is a crude measure of effective sample size,
## and Rhat is the potential scale reduction factor on split chains (at
## convergence, Rhat=1).
## Graphical diagnostics
rats_stan_arr = as.array(rats_stan)
diagnostics(rats_stan_arr)
## NULL
## Rhat values all close to 1, n_eff large, graphical
## diagnostics give no cause for concern.
```

```
/* When using HMC to sample the posterior of hierarchical
  models, it can sometimes improve mixing if we adopt a
  different parameterisation that "decouples" the model
  and top level prior specifications. For the rats example,
  the Stan programme based on the decoupled parameterisation
   is shown below. See Section 15.5 in Gelman et al. (2013)
   for further details. */
data {
 int<lower=0> N;
                      // No. studies (sample size)
 int<lower=0> y[N]; // No. of rats with tumours
 int<lower=0> n[N];
                       // Total number of rats
                        // Prior mean for mu
  real m_mu;
  real<lower=0> s_mu;
                        // Prior std. dev. for mu
  real<lower=0> a_sigsq; // Prior shape for sigsq
  real<lower=0> b_sigsq; // Prior rate for sigsq
}
```




Figure 3: First page of the graphical diagnostics for the rat tumour example.

```
parameters {
  vector[N] phi; // New parameters
  real mu;
  real<lower=0> sigsq;
}
model {
 // Likelihood:
 y ~ binomial_logit(n, mu + sqrt(sigsq) * phi);
  // Prior:
  phi \sim normal(0, 1);
  // "Hyperprior":
  mu ~ normal(m_mu, s_mu);
  sigsq ~ gamma(a_sigsq, b_sigsq);
}
generated quantities {
  vector[N] theta;
  vector<lower=0, upper=1>[N] p; /* Study-specific probability
                                     of tumour */
  real<lower=0, upper=1> p_pop; /* "Population" probability
                                     of tumour */
  theta = mu + sqrt(sigsq) * phi; // Original parameters
  p = inv_logit(theta);
  p_pop = inv_logit(mu);
```

In this example, the decoupled parameterisation offers ## a marginal improvement in mixing ...

```
## ... as illustrated by, e.g.
print(rats_reparam_stan, pars=c("mu", "sigsq", "p_pop"))
## Inference for Stan model: rats_reparam.
## 4 chains, each with iter=2000; warmup=1000; thin=1;
## post-warmup draws per chain=1000, total post-warmup draws=4000.
##
##
         mean se_mean sd 2.5% 25% 50% 75% 97.5% n_eff Rhat
## mu -1.94 0.00 0.12 -2.19 -2.02 -1.93 -1.85 -1.71 2228
## sigsg 0.51 0.01 0.19 0.22 0.38 0.48 0.62 0.94 1331
## p_pop 0.13 0.00 0.01 0.10 0.12 0.13 0.14 0.15 2256
##
## Samples were drawn using NUTS(diag_e) at Tue Apr 4 23:03:54 2017.
## For each parameter, n_eff is a crude measure of effective sample size,
## and Rhat is the potential scale reduction factor on split chains (at
## convergence, Rhat=1).
```

4 Three state mixture model*

In Section 5.4 we fitted a mixture model with K=2 components to the faithful data on the times between successive eruptions of the Old Faithful geyser:

```
data(faithful)
head(faithful)
## eruptions waiting
## 1
      3.600
## 2
       1.800
                54
## 3
      3.333
               74
## 4
      2.283
                62
## 5
      4.533
                85
## 6 2.883
             55
```

In the two-component case, we had component membership probabilities (π_1, π_2) which summed to one and so we reparameterised the model as $\pi_1 = \pi$ and $\pi_2 = 1 - \pi$ where $0 \le \pi \le 1$. We represented this in Stan as a constrained real:

```
real<lower=0,upper=1> pi;
```

In the more general case with K components we have vector $(\pi_1, \pi_2, ..., \pi_K)$ on the K-dimensional simplex. This can be represented in Stan as a simplex vector type:

```
simplex[K] pi_vec;
```

In the case of two-components, we assigned pi a symmetric Beta prior:

```
// Prior:
pi ~ beta(a_pi, a_pi);
```

The multivariate generalisation of this is a symmetric Dirichlet prior.

• Use the lookup function to find the Dirichlet distribution in the Stan manual:

```
lookup("Dirichlet") # Simply search for string rather than
        StanFunction
                                     Arguments ReturnType Page
## 115 dirichlet_lpdf (vector theta | vector alpha) real 505
                                                   real 505
## 116
         dirichlet
## 117 dirichlet_rng
                                (vector alpha) vector 505
              # the name of a particular R function.
```

Try to generalise the Stan programme from Figure 5.7 so that it allows the number K of states to be a component of the list you pass to the stan function through its data argument.

```
/* General K component mixture of normals, saved in
  oldfaithful_gen.stan */
data {
 for pi_vec */
                       // Prior mean for mu[k]
  real m_mu;
 real m_mu; // Prior mean for mu[k]
real<lower=0> s_mu; // Prior std. dev. for mu[k]
  real<lower=0> a_siqsq; // Prior shape for sigsq[k]
  real<lower=0> b_sigsq; // Prior rate for sigsq[k]
}
transformed data {
 vector<lower=0>[K] a_pi_vec;
  for(k in 1:K) a_pi_vec[k] = a_pi;
parameters {
 simplex[K] pi_vec;
 ordered[K] mu;
  vector<lower=0>[K] sigsq;
}
model {
 vector[K] tmp;
 // Likelihood:
  for(n in 1:N) {
   for(k in 1:K) {
     tmp[k] = log(pi_vec[k]) + normal_lpdf(y[n] | mu[k], sqrt(sigsq[k]));
   target += log_sum_exp(tmp);
 }
  // Prior:
  pi_vec ~ dirichlet(a_pi_vec);
 mu ~ normal(m_mu, s_mu);
  sigsq ~ inv_gamma(a_siqsq, b_sigsq);
```

• Test the programme by compiling it then running it with K = 2and K = 3 components.

```
## Set up data to pass to stan function in its "data" argument:
faithful_2_data = list(N=nrow(faithful), y=log(faithful$waiting),
                       K=2, a_pi=3, m_mu=4, s_mu=1,
                       a_siqsq=4, b_sigsq=0.04)
faithful_3_data = faithful_2_data
faithful_3_data[["K"]] = 3
faithful_3_data[["s_mu"]] = 0.1 \# (Try with s_mu=1 and see what
                               # happens ...)
## Compile and run Stan programme:
chains = 4
iter = 2000
thin = 1
faithful_2_stan = stan("oldfaithful_gen.stan", data=faithful_2_data,
                       chains=chains, iter=iter, warmup=iter/2,
                       thin=thin)
iter = 20000
thin = 10
faithful_3_stan = stan("oldfaithful_gen.stan", data=faithful_3_data,
                       chains=chains, iter=iter, warmup=iter/2,
                       thin=thin)
```

```
## Numerical diagnostics for two component model:
print(faithful_2_stan)
## Inference for Stan model: oldfaithful_gen.
## 4 chains, each with iter=2000; warmup=1000; thin=1;
## post-warmup draws per chain=1000, total post-warmup draws=4000.
##
           mean se_mean sd 2.5% 25% 50% 75% 97.5% n_eff
##
## pi_vec[1] 0.38 0.00 0.03 0.31 0.35 0.38 0.40 0.44 2938
## pi_vec[2] 0.62 0.00 0.03 0.56 0.60 0.62 0.65 0.69 2938
## mu[1] 4.00 0.00 0.01 3.98 3.99 4.00 4.01 4.03 1769
## mu[2] 4.38 0.00 0.01 4.37 4.38 4.38 4.39 4.30 4.000
140.50 0.04 1.55 136.78 139.68 140.78 141.65 142.61 1731
## lp__
##
          Rhat
## pi_vec[1] 1
## pi_vec[2] 1
## mu[1]
           1
## mu[2]
            1
## sigsq[1]
            1
## sigsq[2] 1
            1
## lp__
## Samples were drawn using NUTS(diag_e) at Tue Apr 4 23:04:48 2017.
## For each parameter, n_eff is a crude measure of effective sample size,
## and Rhat is the potential scale reduction factor on split chains (at
## convergence, Rhat=1).
```

```
## Graphical diagnostics for two component model:
faithful_2_stan_arr = as.array(faithful_2_stan)
diagnostics(faithful_2_stan_arr)
## NULL
## Rhat values all close to 1, n_eff large, graphical
## diagnostics give no cause for concern.
```




Figure 4: First page of graphical diagnostics for two component mixture.

```
## Numerical diagnostics for three component model:
print(faithful_3_stan)
## Inference for Stan model: oldfaithful_gen.
## 4 chains, each with iter=20000; warmup=10000; thin=10;
## post-warmup draws per chain=1000, total post-warmup draws=4000.
##
##
                                    2.5%
                                            25%
                                                    50%
                                                           75%
                                                                97.5% n_eff
               mean se_mean
                              sd
## pi_vec[1]
               0.20
                       0.00 0.07
                                    0.06
                                           0.15
                                                  0.20
                                                          0.25
                                                                 0.33
                                                                       3579
                       0.00 0.07
## pi_vec[2]
               0.19
                                    0.06
                                           0.14
                                                  0.19
                                                          0.24
                                                                 0.33
                                                                       3621
## pi_vec[3]
               0.61
                       0.00 0.03
                                    0.54
                                           0.59
                                                  0.61
                                                          0.63
                                                                 0.67
                                                                       4000
## mu[1]
               3.96
                       0.00 0.03
                                    3.90
                                           3.94
                                                  3.96
                                                          3.98
                                                                 4.01
                                                                       3437
               4.06
                       0.00 0.04
## mu[2]
                                    4.00
                                           4.03
                                                  4.05
                                                          4.09
                                                                 4.15
                                                                       2545
               4.38
                       0.00 0.01
## mu[3]
                                    4.37
                                           4.38
                                                  4.38
                                                          4.39
                                                                 4.40
                                                                       3978
## sigsq[1]
               0.01
                       0.00 0.00
                                    0.00
                                           0.01
                                                  0.01
                                                          0.01
                                                                 0.02
                                                                       3880
               0.01
                       0.00 0.01
                                           0.01
                                                  0.01
                                                                 0.03
                                                                       3720
## sigsq[2]
                                    0.01
                                                          0.02
## sigsq[3]
               0.01
                       0.00 0.00
                                    0.00
                                           0.00
                                                  0.01
                                                          0.01
                                                                 0.01
                                                                       4000
             136.93
                       0.03 2.13 131.88 135.77 137.24 138.48 140.11 3746
## lp__
```

```
Rhat
## pi_vec[1]
             1
## pi_vec[2]
               1
## pi_vec[3]
## mu[1]
               1
## mu[2]
               1
## mu[3]
              1
## sigsq[1]
## sigsq[2]
              1
              1
## sigsq[3]
## lp__
               1
## Samples were drawn using NUTS(diag_e) at Tue Apr 4 23:09:44 2017.
## For each parameter, n_eff is a crude measure of effective sample size,
## and Rhat is the potential scale reduction factor on split chains (at
## convergence, Rhat=1).
## Graphical diagnostics for three component model:
faithful_3_stan_arr = as.array(faithful_3_stan)
diagnostics(faithful_3_stan_arr)
## NULL
## Numerical and graphical diagnostics do not look
## great, even with lots of iterations and thinning.
## Components 1 and 2 are very similar so I suspect
## there is some overfitting which is affecting the
## performance of the sampler.
```

5 Survival with right censoring*

In this section we will consider another generalised linear model, this time assuming an exponential error distribution. The data we will consider concern the survival times T_n for n = 1, ..., N, in months, of N = 148 renal patients following kidney transplants. There is one covariate x_n for each patient, namely the total number of HLA-B or DR antigen mismatches between the donor and recipient; we might expect survival times to be shorter if there are more mismatches. For some patients, the month of death is observed. However, for others the survival time is *right-censored* meaning we do not observe the time of death, only a time at which the patient was known still to be alive. This can happen for lots of reasons, for example, the study may have ended before the patient died or the patient may have been lost to follow-up. We introduce an indicator variable s_n representing the censoring status of the patient. We set $s_n = 1$ if the corresponding observation t_n on T_n represents the survival time of patient n. On the other hand, we set $s_n = 0$ if the observation t_n on T_n is a right-censored time, meaning all we know is that the survival time of patient *n* is greater than t_n , i.e. $T_n > t_n$.

The renal data set in the jrRstan package contains the survival data:




Figure 5: First page of graphical diagnostics for three component mixture.

Our model for the survival times can be expressed as:

$$T_n|\lambda_n \sim \operatorname{Exp}(\lambda_n)$$
, independently for $n = 1, 2, \dots, N$,

where $\lambda_n = \exp(\eta_n)$ is the reciprocal of the mean survival time for patient n. The corresponding *linear predictor* η_n takes the form:

$$\eta_n = \beta_1 + \beta_2 x_n.$$

We adopt the following prior for the two model parameters:

$$\beta_1 \sim N(0,30^2)$$
, and, independently, $\beta_2 \sim N(0,30^2)$.

If all the survival times were observed, i.e. if $s_n = 1$ for all n = 1, ..., N, the likelihood would take the form:

$$\prod_{n=1}^{N} p(t_n|\lambda_n)$$

in which patient n contributes $p(t_n|\lambda_n)$ to the likelihood function, namely the density function of the $\operatorname{Exp}(\lambda_n)$ distribution evaluated at the observed survival time $T_n = t_n$. However, if the time for patient n is right-censored our observation t_n does not represent the survival time T_n , and we only know that $T_n > t_n$. For such patients the contribution to the likelihood is therefore $\operatorname{Pr}(T_n > t_n|\lambda_n)$ which is the survival function, or complementary cumulative distribution function, of the $\operatorname{Exp}(\lambda_n)$ distribution evaluated at the censored time t_n . Overall, therefore, the likelihood takes the form

$$\prod_{n:s_n=1} p(t_n|\lambda_n) \times \prod_{n:s_n=0} \Pr(T_n > t_n|\lambda_n)$$

where the first term is a product over the patients for whom we observe the survival times and the second term is a product over the patients whose survival times are right-censored. Because the Stan modelling language is very flexible, we can handle the non-standard second term in the likelihood by incrementing the log posterior density using the target keyword⁴ and the log complementary cumulative distribution function of the exponential distribution, exponential_lccdf.

⁴ As in the mixture model example.

• Write a Stan model to represent the model and prior above.

```
// Survival model with right-censoring, saved in renal_intout.stan
data {
 int<lower=0> N_obs;
                             /* No. patients with observed
                                survival times */
 int<lower=0> N_cens; /* No. patients with right-censored
                              times */
                             // No. columns in design matrix
 int<lower=1> J;
 vector<lower=0>[N_obs] t_obs; // Observed survival times
 vector<lower=0>[N_cens] t_cens; // Right-censored times
 matrix[N_obs, J] X_obs; /* Design matrix for observed
                                patients */
 matrix[N_cens, J] X_cens; /* Design matrix for
                                right-censored patients */
                            // Means in beta prior
  real beta_mean[J];
 }
parameters {
 vector[J] beta;
}
model {
 // Likelihood for observed patients:
 t_obs ~ exponential(exp(X_obs * beta));
 // Likelihood for right-censored patients:
 target += exponential_lccdf(t_cens | exp(X_cens * beta));
 // Prior:
 beta ~ normal(beta_mean, beta_sd);
```

• Create a suitable data representation in R then compile and run the Stan programme.

```
## Set up data to pass to stan function in its "data" argument:
N_obs = nrow(renal[renal$status==1,]) # No. observed patients
N_cens = nrow(renal[renal$status==0,]) # No. right-censored patients
X_{obs} = matrix(1, N_{obs}, 2)
                                      # Design matrix for observed
X_obs[,2] = renal$x[renal$status==1] # patients
X_{cens} = matrix(1, N_{cens}, 2)
                                 # Design matrix for
X_cens[,2] = renal$x[renal$status==0] # right-censored patients
renal_data = list(N_obs=N_obs, N_cens=N_cens,
                 t_obs = renal$t[renal$status==1],
                  t_cens = renal$t[renal$status==0],
                 X_obs=X_obs, X_cens=X_cens, J=2,
                 beta_mean = c(0, 0), beta_sd = c(30, 30))
## Compile and run Stan programme:
chains = 4
iter = 10000
thin = 5
renal_stan = stan("renal_intout.stan", data=renal_data,
                   chains=chains, iter=iter, warmup=iter/2,
                   thin=thin)
```

• Check both numerical and graphical diagnostics.

```
## Numerical diagnostics:
print(renal_stan)
## Inference for Stan model: renal_intout.
## 4 chains, each with iter=10000; warmup=5000; thin=5;
## post-warmup draws per chain=1000, total post-warmup draws=4000.
##
           mean se_mean sd 2.5% 25%
                                               50%
                                                       75% 97.5% n_eff
## beta[1] -5.97 0.01 0.40 -6.80 -6.23 -5.95 -5.70 -5.25 3244
## beta[2] 0.98 0.00 0.17
                              0.65 0.87 0.98 1.09
                                                             1.30 3279
         -152.48
## lp__
                  0.02 1.02 -155.20 -152.89 -152.16 -151.75 -151.49 3275
##
         Rhat
## beta[1] 1
## beta[2] 1
## lp__
##
## Samples were drawn using NUTS(diag_e) at Tue Apr \, 4 23:10:32 2017.
## For each parameter, n_eff is a crude measure of effective sample size,
## and Rhat is the potential scale reduction factor on split chains (at
## convergence, Rhat=1).
## Graphical diagnostics:
renal_stan_arr = as.array(renal_stan)
diagnostics(renal_stan_arr)
## NULL
## Rhat values all close to 1, n_eff large, graphical
## diagnostics give no cause for concern.
```




Figure 6: Graphical diagnostics for the survival model with right-censoring.

Solutions

Solutions are available as a vignette:

```
library("jrRstan")
vignette("solutions2", package="jrRstan")
```